
Solving a system of polynomial equations is a ubiquitous problem in the
applications of mathematics. Until recently, it has been hopeless to find ex-
plicit solutions to such systems, and mathematics has instead developed deep
and powerful theories about the solutions to polynomial equations. Enumer-
ative Geometry is concerned with counting the number of solutions when the
polynomials come from a geometric situation and Intersection Theory gives
methods to accomplish the enumeration.

We use Macaulay 2 to investigate some problems from enumerative geom-
etry, illustrating some applications of symbolic computation to this important
problem of solving systems of polynomial equations. Besides enumerating so-
lutions to the resulting polynomial systems, which include overdetermined,
deficient, and improper systems, we address the important question of real
solutions to these geometric problems.

1 Introduction

A basic question to ask about a system of polynomial equations is its number
of solutions. For this, the fundamental result is the following Bézout Theorem.

Theorem 1.1. The number of isolated solutions to a system of polynomial
equations

f1(x1, . . . , xn) = f2(x1, . . . , xn) = · · · = fn(x1, . . . , xn) = 0

is bounded by d1d2 · · · dn, where di := deg fi. If the polynomials are generic,
then this bound is attained for solutions in an algebraically closed field.

Here, isolated is taken with respect to the algebraic closure. This Bézout
Theorem is a consequence of the refined Bézout Theorem of Fulton and
MacPherson [12, §1.23].

A system of polynomial equations with fewer than this degree bound or
Bézout number of solutions is called deficient, and there are well-defined
classes of deficient systems that satisfy other bounds. For example, fewer
monomials lead to fewer solutions, for which polyhedral bounds [4] on the
number of solutions are often tighter (and no weaker than) the Bézout num-
ber, which applies when all monomials are present. When the polynomials
? Supported in part by NSF grant DMS-0070494.

come from geometry, determining the number of solutions is the central prob-
lem in enumerative geometry.

Symbolic computation can help compute the solutions to a system of equa-
tions that has only isolated solutions. In this case, the polynomials generate
a zero-dimensional ideal I. The degree of I is dimk k[X]/I, the dimension of
the k-vector space k[X]/I, which is also the number of standard monomi-
als in any term order. This degree gives an upper bound on the number of
solutions, which is attained when I is radical.

Example 1.2. We illustrate this discussion with an example. Let f1, f2, f3,
and f4 be random quadratic polynomials in the ring F101[y11, y12, y21, y22].

i1 : R = ZZ/101[y11, y12, y21, y22];

i2 : PolynomialSystem = apply(1..4, i ->
random(0, R) + random(1, R) + random(2, R));

The ideal they generate has dimension 0 and degree 16 = 24, which is the
Bézout number.

i3 : I = ideal PolynomialSystem;

o3 : Ideal of R

i4 : dim I, degree I

o4 = (0, 16)

o4 : Sequence

If we restrict the monomials which appear in the fi to be among

1, y11, y12, y21, y22, y11y22, and y12y21,

then the ideal they generate again has dimension 0, but its degree is now 4.
i5 : J = ideal (random(R^4, R^7) * transpose(

matrix{{1, y11, y12, y21, y22, y11*y22, y12*y21}}));

o5 : Ideal of R

i6 : dim J, degree J

o6 = (0, 4)

o6 : Sequence

If we further require that the coefficients of the quadratic terms sum to zero,
then the ideal they generate now has degree 2.

i7 : K = ideal (random(R^4, R^6) * transpose(
matrix{{1, y11, y12, y21, y22, y11*y22 - y12*y21}}));

o7 : Ideal of R

i8 : dim K, degree K

o8 = (0, 2)

o8 : Sequence

In Example 4.2, we shall see how this last specialization is geometrically
meaningful.

For us, enumerative geometry is concerned with enumerating geometric
figures of some kind having specified positions with respect to general fixed
figures. That is, counting the solutions to a geometrically meaningful system
of polynomial equations. We use Macaulay 2 to investigate some enumerative
geometric problems from this point of view. The problem of enumeration will
be solved by computing the degree of the (0-dimensional) ideal generated by
the polynomials.

2 Solving Systems of Polynomials

We briefly discuss some aspects of solving systems of polynomial equations.
For a more complete survey, see the relevant chapters in [6,7].

Given an ideal I in a polynomial ring k[X], set V(I) := Spec k[X]/I.
When I is generated by the polynomials f1, . . . , fN , V(I) gives the set of
solutions in affine space to the system

f1(X) = · · · = fN (X) = 0 (1)

a geometric structure. These solutions are the roots of the ideal I. The degree
of a zero-dimensional ideal I provides an algebraic count of its roots. The de-
gree of its radical counts roots in the algebraic closure, ignoring multiplicities.

2.1 Excess Intersection

Sometimes, only a proper (open) subset of affine space is geometrically mean-
ingful, and we want to count only the meaningful roots of I. Often the roots
V(I) has positive dimensional components that lie in the complement of the
meaningful subset. One way to treat this situation of excess or improper in-
tersection is to saturate I by a polynomial f vanishing on the extraneous
roots. This has the effect of working in k[X][f−1], the coordinate ring of the
complement of V(f) [9, Exer. 2.3].

Example 2.1. We illustrate this with an example. Consider the following
ideal in F7[x, y].

i9 : R = ZZ/7[y, x, MonomialOrder=>Lex];

i10 : I = ideal (y^3*x^2 + 2*y^2*x + 3*x*y, 3*y^2 + x*y - 3*y);

o10 : Ideal of R

Since the generators have greatest common factor y, I defines finitely many
points together with the line y = 0. Saturate I by the variable y to obtain
the ideal J of isolated roots.

i11 : J = saturate(I, ideal(y))

4 3 2
o11 = ideal (x + x + 3x + 3x, y - 2x - 1)

o11 : Ideal of R

The first polynomial factors completely in F7[x],
i12 : factor(J_0)

o12 = (x)(x - 2)(x + 2)(x + 1)

o12 : Product

and so the isolated roots of I are (2, 5), (−1,−1), (0, 1), and (−2,−3).

Here, the extraneous roots came from a common factor in both equations.
A less trivial example of this phenomenon will be seen in Section 5.2.

2.2 Elimination, Rationality, and Solving

Elimination theory can be used to study the roots of a zero-dimensional ideal
I ⊂ k[X]. A polynomial h ∈ k[X] defines a map k[y]→ k[X] (by y 7→ h) and
a corresponding projection h : Spec k[X] � A

1. The generator g(y) ∈ k[y]
of the kernel of the map k[y] → k[X]/I is called an eliminant and it has
the property that V(g) = h(V(I)). When h is a coordinate function xi, we
may consider the eliminant to be in the polynomial ring k[xi], and we have
〈g(xi)〉 = I ∩ k[xi]. The most important result concerning eliminants is the
Shape Lemma [2].

Shape Lemma. Suppose h is a linear polynomial and g is the corresponding
eliminant of a zero-dimensional ideal I ⊂ k[X] with deg(I) = deg(g). Then
the roots of I are defined in the splitting field of g and I is radical if and only
if g is square-free.

Suppose further that h = x1 so that g = g(x1). Then, in the lexicographic
term order with x1 < x2 < · · · < xn, I has a Gröbner basis of the form:

g(x1), x2 − g2(x1), . . . , xn − gn(x1) , (2)

where deg(g) > deg(gi) for i = 2, . . . , n.

When k is infinite and I is radical, an eliminant g given by a generic linear
polynomial h will satisfy deg(g) = deg(I). Enumerative geometry counts
solutions when the fixed figures are generic. We are similarly concerned with
the generic situation of deg(g) = deg(I). In this case, eliminants provide a
useful computational device to study further questions about the roots of I.
For instance, the Shape Lemma holds for the saturated ideal of Example 2.1.
Its eliminant, which is the polynomial J_0, factors completely over the ground
field F7, so all four solutions are defined in F7. In Section 4.3, we will use
eliminants in another way, to show that an ideal is radical.



From Enumerative Geometry to Solving Equations 5

Given a polynomial h in a zero-dimensional ring k[X]/I, the procedure
eliminant(h, k[y]) finds a linear relation modulo I among the powers
1, h, h2, . . . , hd of h with d minimal and returns this as a polynomial in k[y].
This procedure is included in the Macaulay 2 package realroots.m2.

i13 : load "realroots.m2"

i14 : code eliminant

o14 = -- realroots.m2:65-80
eliminant = (h, C) -> (

Z := C_0;
A := ring h;
assert( dim A == 0 );
F := coefficientRing A;
assert( isField F );
assert( F == coefficientRing C );
B := basis A;
d := numgens source B;
M := fold((M, i) -> M ||

substitute(contract(B, h^(i+1)), F),
substitute(contract(B, 1_A), F),
flatten subsets(d, d));

N := ((ker transpose M)).generators;
P := matrix {toList apply(0..d, i -> Z^i)} * N;

(flatten entries(P))_0)

Here, M is a matrix whose rows are the normal forms of the powers 1, h,
h2, . . ., hd of h, for d the degree of the ideal. The columns of the kernel
N of transpose M are a basis of the linear relations among these powers.
The matrix P converts these relations into polynomials. Since N is in column
echelon form, the initial entry of P is the relation of minimal degree. (This
method is often faster than näıvely computing the kernel of the map k[Z]→ A
given by Z 7→ h, which is implemented by eliminantNaive(h, Z).)

Suppose we have an eliminant g(x1) of a zero-dimensional ideal I ⊂ k[X]
with deg(g) = deg(I), and we have computed the lexicographic Gröbner
basis (2). Then the roots of I are

{(ξ1, g2(ξ1), . . . , gn(ξ1)) | g(ξ1) = 0} . (3)

Suppose now that k = Q and we seek floating point approximations for
the (complex) roots of I. Following this method, we first compute floating
point solutions to g(ξ) = 0, which give all the x1-coordinates of the roots of
I, and then use (3) to find the other coordinates. The difficulty here is that
enough precision may be lost in evaluating gi(ξ1) so that the result is a poor
approximation for the other components ξi.

2.3 Solving with Linear Algebra

We describe another method based upon numerical linear algebra. When
I ⊂ k[X] is zero-dimensional, A = k[X]/I is a finite-dimensional k-vector
space, and any Gröbner basis for I gives an efficient algorithm to compute

ring operations using linear algebra. In particular, multiplication by h ∈ A is
a linear transformation mh : A→ A and the command regularRep(h) from
realroots.m2 gives the matrix of mh in terms of the standard basis of A.

i15 : code regularRep

o15 = -- realroots.m2:96-100
regularRep = f -> (

assert( dim ring f == 0 );
b := basis ring f;
k := coefficientRing ring f;
substitute(contract(transpose b, f*b), k))

Since the action of A on itself is faithful, the minimal polynomial of
mh is the eliminant corresponding to h. The procedure charPoly(h, Z) in
realroots.m2 computes the characteristic polynomial det(Z · Id−mh) of h.

i16 : code charPoly

o16 = -- realroots.m2:106-113
charPoly = (h, Z) -> (

A := ring h;
F := coefficientRing A;
S := F[Z];
Z = value Z;
mh := regularRep(h) ** S;
Idz := S_0 * id_(S^(numgens source mh));
det(Idz - mh))

When this is the minimal polynomial (the situation of the Shape Lemma),
this procedure often computes the eliminant faster than does eliminant,
and for systems of moderate degree, much faster than näıvely computing the
kernel of the map k[Z]→ A given by Z 7→ h.

The eigenvalues and eigenvectors of mh give another algorithm for finding
the roots of I. The engine for this is the following result.

Stickelberger’s Theorem. Let h ∈ A and mh be as above. Then there is
a one-to-one correspondence between eigenvectors vξ of mh and roots ξ of I,
the eigenvalue of mh on vξ is the value h(ξ) of h at ξ, and the multiplicity
of this eigenvalue (on the eigenvector vξ) is the multiplicity of the root ξ.

Since the linear transformations mh for h ∈ A commute, the eigenvec-
tors vξ are common to all mh. Thus we may compute the roots of a zero-
dimensional ideal I ⊂ k[X] by first computing floating-point approximations
to the eigenvectors vξ of mx1 . Then the root ξ = (ξ1, . . . , ξn) of I corre-
sponding to the eigenvector vξ has ith coordinate satisfying

mxi · vξ = ξi · vξ . (4)

An advantage of this method is that we may use structured numerical lin-
ear algebra after the matrices mxi are precomputed using exact arithmetic.
(These matrices are typically sparse and have additional structures which
may be exploited.) Also, the coordinates ξi are linear functions of the float-
ing point entries of vξ, which affords greater precision than the non-linear

evaluations gi(ξ1) in the method based upon elimination. While in principle
only one of the deg(I) components of the vectors in (4) need be computed,
averaging the results from all components can improve precision.

2.4 Real Roots

Determining the real roots of a polynomial system is a challenging problem
with real world applications. When the polynomials come from geometry, this
is the main problem of real enumerative geometry. Suppose k ⊂ R and I ⊂
k[X] is zero-dimensional. If g is an eliminant of k[X]/I with deg(g) = deg(I),
then the real roots of g are in 1-1 correspondence with the real roots of I.
Since there are effective methods for counting the real roots of a univariate
polynomial, eliminants give a näıve, but useful method for determining the
number of real roots to a polynomial system. (For some applications of this
technique in mathematics, see [20,23,25].)

The classical symbolic method of Sturm, based upon Sturm sequences,
counts the number of real roots of a univariate polynomial in an interval.
When applied to an eliminant satisfying the Shape Lemma, this method
counts the number of real roots of the ideal. This is implemented in Macau-
lay 2 via the command SturmSequence(f) of realroots.m2

i17 : code SturmSequence

o17 = -- realroots.m2:117-131
SturmSequence = f -> (

assert( isPolynomialRing ring f );
assert( numgens ring f === 1 );
R := ring f;
assert( char R == 0 );
x := R_0;
n := first degree f;
c := new MutableList from toList (0 .. n);
if n >= 0 then (

c#0 = f;
if n >= 1 then (

c#1 = diff(x,f);
scan(2 .. n, i -> c#i = - c#(i-2) % c#(i-1));
));

toList c)

The last few lines of SturmSequence construct the Sturm sequence of the
univariate argument f : This is (f0, f1, f2, . . .) where f0 = f , f1 = f ′, and for
i > 1, fi is the normal form reduction of −fi−2 modulo fi−1. Given any real
number x, the variation of f at x is the number of changes in sign of the
sequence (f0(x), f1(x), f2(x), . . .) obtained by evaluating the Sturm sequence
of f at x. Then the number of real roots of f over an interval [x, y] is the
difference of the variation of f at x and at y.

The Macaulay 2 commands numRealSturm and numPosRoots (and also
numNegRoots) use this method to respectively compute the total number of
real roots and the number of positive roots of a univariate polynomial.

i18 : code numRealSturm

o18 = -- realroots.m2:160-163
numRealSturm = f -> (

c := SturmSequence f;
variations (signAtMinusInfinity \ c)

- variations (signAtInfinity \ c))

i19 : code numPosRoots

o19 = -- realroots.m2:168-171
numPosRoots = f -> (

c := SturmSequence f;
variations (signAtZero \ c)

- variations (signAtInfinity \ c))

These use the commands signAt∗(f), which give the sign of f at ∗. (Here, ∗
is one of Infinity, zero, or MinusInfinity.) Also variations(c) computes
the number of sign changes in the sequence c.

i20 : code variations

o20 = -- realroots.m2:183-191
variations = c -> (

n := 0;
last := 0;
scan(c, x -> if x != 0 then (

if last < 0 and x > 0 or last > 0
and x < 0 then n = n+1;

last = x;
));

n)

A more sophisticated method to compute the number of real roots which
can also give information about their location uses the rank and signature
of the symmetric trace form. Suppose I ⊂ k[X] is a zero-dimensional ideal
and set A := k[X]/I. For h ∈ k[X], set Sh(f, g) := trace(mhfg). It is an
easy exercise that Sh is a symmetric bilinear form on A. The procedure
traceForm(h) in realroots.m2 computes this trace form Sh.

i21 : code traceForm

o21 = -- realroots.m2:196-203
traceForm = h -> (

assert( dim ring h == 0 );
b := basis ring h;
k := coefficientRing ring h;
mm := substitute(contract(transpose b, h * b ** b), k);
tr := matrix {apply(first entries b, x ->

trace regularRep x)};
adjoint(tr * mm, source tr, source tr))

The value of this construction is the following theorem.

Theorem 2.2 ([3,19]). Suppose k ⊂ R and I is a zero-dimensional ideal
in k[x1, . . . , xn] and consider V(I) ⊂ Cn. Then, for h ∈ k[x1, . . . , xn], the
signature σ(Sh) and rank ρ(Sh) of the bilinear form Sh satisfy

σ(Sh) = #{a ∈ V(I) ∩ Rn : h(a) > 0} −#{a ∈ V(I) ∩ Rn : h(a) < 0}
ρ(Sh) = #{a ∈ V(I) : h(a) 6= 0} .

That is, the rank of Sh counts roots in Cn−V(h), and its signature counts
the real roots weighted by the sign of h (which is −1, 0, or 1) at each root.
The command traceFormSignature(h) in realroots.m2 returns the rank
and signature of the trace form Sh.

i22 : code traceFormSignature

o22 = -- realroots.m2:208-218
traceFormSignature = h -> (

A := ring h;
assert( dim A == 0 );
assert( char A == 0 );
S := QQ[Z];
TrF := traceForm(h) ** S;
IdZ := Z * id_(S^(numgens source TrF));
f := det(TrF - IdZ);
<< "The trace form S_h with h = " << h <<

" has rank " << rank(TrF) << " and signature " <<
numPosRoots(f) - numNegRoots(f) << endl; )

The Macaulay 2 command numRealTrace(A) simply returns the number of
real roots of I, given A = k[X]/I.

i23 : code numRealTrace

o23 = -- realroots.m2:223-230
numRealTrace = A -> (

assert( dim A == 0 );
assert( char A == 0 );
S := QQ[Z];
TrF := traceForm(1_A) ** S;
IdZ := Z * id_(S^(numgens source TrF));
f := det(TrF - IdZ);
numPosRoots(f)-numNegRoots(f))

Example 2.3. We illustrate these methods on the following polynomial sys-
tem.

i24 : R = QQ[x, y];

i25 : I = ideal (1 - x^2*y + 2*x*y^2, y - 2*x - x*y + x^2);

o25 : Ideal of R

The ideal I has dimension zero and degree 5.
i26 : dim I, degree I

o26 = (0, 5)

o26 : Sequence

We compare the two methods to compute the eliminant of x in the ring R/I .
i27 : A = R/I;

i28 : time g = eliminant(x, QQ[Z])
-- used 0.09 seconds

5 4 3 2
o28 = Z - 5Z + 6Z + Z - 2Z + 1

o28 : QQ [Z]

i29 : time g = charPoly(x, Z)
-- used 0.02 seconds

5 4 3 2
o29 = Z - 5Z + 6Z + Z - 2Z + 1

o29 : QQ [Z]

The eliminant has 3 real roots, which we test in two different ways.
i30 : numRealSturm(g), numRealTrace(A)

o30 = (3, 3)

o30 : Sequence

We use Theorem 2.2 to isolate these roots in the x, y-plane.
i31 : traceFormSignature(x*y);
The trace form S_h with h = x*y has rank 5 and signature 3

Thus all 3 real roots lie in the first and third quadrants (where xy > 0). We
isolate these further.

i32 : traceFormSignature(x - 2);
The trace form S_h with h = x - 2 has rank 5 and signature 1

This shows that two roots lie in the first quadrant with x > 2 and one lies
in the third. Finally, one of the roots lies in the triangle y > 0, x > 2, and
x+ y < 3.

i33 : traceFormSignature(x + y - 3);
The trace form S_h with h = x + y - 3 has rank 5 and signature -1

Figure 1 shows these three roots (dots), as well as the lines x+ y = 3 and
x = 2.

Fig. 1. Location of roots

2.5 Homotopy Methods

We describe symbolic-numeric homotopy continuation methods for finding
approximate complex solutions to a system of equations. These exploit the
traditional principles of conservation of number and specialization from enu-
merative geometry.

Suppose we seek the isolated solutions of a system F (X) = 0 where
F = (f1, . . . , fn) are polynomials in the variables X = (x1, . . . , xN ). First, a
homotopy H(X, t) is found with the following properties:

1. H(X, 1) = F (X).
2. The isolated solutions of the start system H(X, 0) = 0 are known.
3. The system H(X, t) = 0 defines finitely many (complex) curves, and

each isolated solution of the original system F (X) = 0 is connected to an
isolated solution σi(0) of H(X, 0) = 0 along one of these curves.

Next, choose a generic smooth path γ(t) from 0 to 1 in the complex plane.
Lifting γ to the curves H(X, t) = 0 gives smooth paths σi(t) connecting each
solution σi(0) of the start system to a solution of the original system. The
path γ must avoid the finitely many points in C over which the curves are
singular or meet other components of the solution set H(X, t) = 0.

Numerical path continuation is used to trace each path σi(t) from t = 0
to t = 1. When there are fewer solutions to F (X) = 0 than to H(X, 0) = 0,
some paths will diverge or become singular as t → 1, and it is expensive to
trace such a path. The homotopy is optimal when this does not occur.

When N = n and the fi are generic, set G(X) := (g1, . . . , gn) with gi =
(xi − 1)(xi − 2) · · · (xi − di) where di := deg(fi). Then the Bézout homotopy

H(X, t) := tF (X) + (1− t)G(X)

is optimal. This homotopy furnishes an effective demonstration of the bound
in Bézout’s Theorem for the number of solutions to F (X) = 0.

When the polynomial system is deficient, the Bézout homotopy is not op-
timal. When n > N (often the case in geometric examples), the Bézout ho-
motopy does not apply. In either case, a different strategy is needed. Present
optimal homotopies for such systems all exploit some structure of the systems
they are designed to solve. The current state-of-the-art is described in [29].

Example 2.4. The Gröbner homotopy [14] is an optimal homotopy that
exploits a square-free initial ideal. Suppose our system has the form

F := g1(X), . . . , gm(X), Λ1(X), . . . , Λd(X)

where g1(X), . . . , gm(X) form a Gröbner basis for an ideal I with respect
to a given term order ≺, Λ1, . . . , Λd are linear forms with d = dim(V(I)),
and we assume that the initial ideal in≺I is square-free. This last, restrictive,
hypothesis occurs for certain determinantal varieties.

As in [9, Chapter 15], there exist polynomials gi(X, t) interpolating be-
tween gi(X) and their initial terms in≺gi(X)

gi(X; 1) = gi(X) and gi(X; 0) = in≺gi(X)

so that 〈g1(X, t), . . . , gm(X, t)〉 is a flat family with generic fibre isomorphic
to I and special fibre in≺I. The Gröbner homotopy is

H(X, t) := g1(X, t), . . . , gm(X, t), Λ1(X), . . . , Λd(X).



12 F. Sottile

Since in≺I is square-free, V(in≺I) is a union of deg(I)-many coordinate d-
planes. We solve the start system by linear algebra. This conceptually simple
homotopy is in general not efficient as it is typically overdetermined.

3 Some Enumerative Geometry

We use the tools we have developed to explore the enumerative geometric
problems of cylinders meeting 5 general points and lines tangent to 4 spheres.

3.1 Cylinders Meeting 5 Points

A cylinder is the locus of points equidistant from a fixed line in R3. The
Grassmannian of lines in 3-space is 4-dimensional, which implies that the
space of cylinders is 5-dimensional, and so we expect that 5 points in R3 will
determine finitely many cylinders. That is, there should be finitely many lines
equidistant from 5 general points. The question is: How many cylinders/lines,
and how many of them can be real?

Bottema and Veldkamp [5] show there are 6 complex cylinders and Licht-
blau [17] observes that if the 5 points are the vertices of a bipyramid con-
sisting of 2 regular tetrahedra sharing a common face, then all 6 will be real.
We check this reality on a configuration with less symmetry (so the Shape
Lemma holds).

If the axial line has direction V and contains the point P (and hence has
parameterization P + tV), and if r is the squared radius, then the cylinder
is the set of points X satisfying

0 = r −
∥∥∥∥X−P− V · (X−P)

‖V‖2
V
∥∥∥∥2

.

Expanding and clearing the denominator of ‖V‖2 yields

0 = r‖V‖2 + [V · (X−P)]2 − ‖X−P‖2 ‖V‖2 . (5)

We consider cylinders containing the following 5 points, which form an asym-
metric bipyramid.

i34 : Points = {{2, 2, 0 }, {1, -2, 0}, {-3, 0, 0},
{0, 0, 5/2}, {0, 0, -3}};

Suppose that P = (0, y11, y12) and V = (1, y21, y22).
i35 : R = QQ[r, y11, y12, y21, y22];

i36 : P = matrix{{0, y11, y12}};

1 3
o36 : Matrix R <--- R

i37 : V = matrix{{1, y21, y22}};

1 3
o37 : Matrix R <--- R

We construct the ideal given by evaluating the polynomial (5) at each of the
five points.

i38 : Points = matrix Points ** R;

5 3
o38 : Matrix R <--- R

i39 : I = ideal apply(0..4, i -> (
X := Points^{i};
r * (V * transpose V) +
((X - P) * transpose V)^2) -
((X - P) * transpose(X - P)) * (V * transpose V)

);

o39 : Ideal of R

This ideal has dimension 0 and degree 6.
i40 : dim I, degree I

o40 = (0, 6)

o40 : Sequence

There are 6 real roots, and they correspond to real cylinders (with r > 0).
i41 : A = R/I; numPosRoots(charPoly(r, Z))

o42 = 6

3.2 Lines Tangent to 4 Spheres

We now ask for the lines having a fixed distance from 4 general points. Equiv-
alently, these are the lines mutually tangent to 4 spheres of equal radius.
Since the Grassmannian of lines is four-dimensional, we expect there to be
only finitely many such lines. Macdonald, Pach, and Theobald [18] show that
there are indeed 12 lines, and that all 12 may be real. This problem makes
geometric sense over any field k not of characteristic 2, and the derivation of
the number 12 is also valid for algebraically closed fields not of characteristic
2.

A sphere in k3 is given by V(q(1,x)), where q is some quadratic form on
k4. Here x ∈ k3 and we note that not all quadratic forms give spheres. If our
field does not have characteristic 2, then there is a symmetric 4 × 4 matrix
M such that q(u) = uMut.

A line ` having direction V and containing the point P is tangent to the
sphere defined by q when the univariate polynomial in s

q((1,P) + s(0,V)) = q(1,P) + 2s(1,P)M(0,V)t + s2q(0,V) ,

has a double root. Thus its discriminant vanishes, giving the equation(
(1,P)M(0,V)t

)2 − (1,P)M(1,P)t · (0,V)M(0,V)t = 0 . (6)

The matrix M of the quadratic form q of the sphere with center (a, b, c)
and squared radius r is constructed by Sphere(a,b,c,r).

i43 : Sphere = (a, b, c, r) -> (
matrix{{a^2 + b^2 + c^2 - r ,-a ,-b ,-c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ -c , 0 , 0 , 1 }}

);

If a line ` contains the point P = (0, y11, y12) and ` has direction V =
(1, y21, y22), then tangentTo(M) is the equation for ` to be tangent to the
quadric uMuT = 0 determined by the matrix M .

i44 : R = QQ[y11, y12, y21, y22];

i45 : tangentTo = (M) -> (
P := matrix{{1, 0, y11, y12}};
V := matrix{{0, 1, y21, y22}};
(P * M * transpose V)^2 -

(P * M * transpose P) * (V * M * transpose V)
);

The ideal of lines having distance
√

5 from the four points (0, 0, 0), (4, 1, 1),
(1, 4, 1), and (1, 1, 4) has dimension zero and degree 12.

i46 : I = ideal (tangentTo(Sphere(0,0,0,5)),
tangentTo(Sphere(4,1,1,5)),
tangentTo(Sphere(1,4,1,5)),
tangentTo(Sphere(1,1,4,5)));

o46 : Ideal of R

i47 : dim I, degree I

o47 = (0, 12)

o47 : Sequence

Thus there are 12 lines whose distance from those 4 points is
√

5. We check
that all 12 are real.

i48 : A = R/I;

i49 : numRealSturm(eliminant(y11 - y12 + y21 + y22, QQ[Z]))

o49 = 12

Since no eliminant given by a coordinate function satisfies the hypotheses
of the Shape Lemma, we took the eliminant with respect to the linear form
y11 − y12 + y21 + y22.

This example is an instance of Lemma 3 of [18]. These four points define a
regular tetrahedron with volume V = 9 where each face has area A =

√
35/2

and each edge has length e =
√

18. That result guarantees that all 12 lines
will be real when e/2 < r < A2/3V , which is the case above.

4 Schubert Calculus

The classical Schubert calculus of enumerative geometry concerns linear sub-
spaces having specified positions with respect to other, fixed subspaces. For
instance, how many lines in P3 meet four given lines? (See Example 4.2.)

More generally, let 1 < r < n and suppose that we are given general
linear subspaces L1, . . . , Lm of kn with dimLi = n − r + 1 − li. When
l1 + · · ·+ lm = r(n− r), there will be a finite number d(r, n; l1, . . . , lm) of r-
planes in kn which meet each Li non-trivially. This number may be computed
using classical algorithms of Schubert and Pieri (see [16]).

The condition on r-planes to meet a fixed (n−r+1−l)-plane non-trivially
is called a (special) Schubert condition, and we call the data (r, n; l1, . . . , lm)
(special) Schubert data. The (special) Schubert calculus concerns this class of
enumerative problems. We give two polynomial formulations of this special
Schubert calculus, consider their solutions over R, and end with a question
for fields of arbitrary characteristic.

4.1 Equations for the Grassmannian

The ambient space for the Schubert calculus is the Grassmannian of r-planes
in kn, denoted Gr,n. For H ∈ Gr,n, the rth exterior product of the embedding
H → kn gives a line

k ' ∧rH −→ ∧rkn ' k(nr) .

This induces the Plücker embedding Gr,n ↪→ P
(nr)−1. If H is the row space

of an r by n matrix, also written H, then the Plücker embedding sends H
to its vector of

(
n
r

)
maximal minors. Thus the r-subsets of {0, . . . , n−1},

Yr,n := subsets(n, r), index Plücker coordinates of Gr,n. The Plücker ideal
of Gr,n is therefore the ideal of algebraic relations among the maximal minors
of a generic r by n matrix.

We create the coordinate ring k[pα | α ∈ Y2,5] of P9 and the Plücker ideal
of G2,5. The Grassmannian Gr,n of r-dimensional subspaces of kn is also the
Grassmannian of r−1-dimensional affine subspaces of Pn−1. Macaulay 2 uses
this alternative indexing scheme.

i50 : R = ZZ/101[apply(subsets(5,2), i -> p_i )];

i51 : I = Grassmannian(1, 4, R)

o51 = ideal (p p - p p + p p , p · · ·
{2, 3} {1, 4} {1, 3} {2, 4} {1, 2} {3, 4} {2, 3} · · ·

o51 : Ideal of R

This projective variety has dimension 6 and degree 5
i52 : dim(Proj(R/I)), degree(I)

o52 = (6, 5)

o52 : Sequence

This ideal has an important combinatorial structure [28, Example 11.9].
We write each α ∈ Yr,n as an increasing sequence α : α1 < · · · < αr. Given
α, β ∈ Yr,n, consider the two-rowed array with α written above β. We say
α ≤ β if each column weakly increases. If we sort the columns of an array

with rows α and β, then the first row is the meet α∧β (greatest lower bound)
and the second row the join α ∨ β (least upper bound) of α and β. These
definitions endow Yr,n with the structure of a distributive lattice. Figure 2
shows Y2,5.

Fig. 2. Y2,5

We give k[pα] the degree reverse lexicographic order, where we first order
the variables pα by lexicographic order on their indices α.

Theorem 4.1. The reduced Gröbner basis of the Plücker ideal with respect to
this degree reverse lexicographic term order consists of quadratic polynomials

g(α, β) = pα · pβ − pα∨β · pα∧β + lower terms in ≺ ,

for each incomparable pair α, β in Yr,n, and all lower terms λpγ ·pδ in g(α, β)
satisfy γ ≤ α ∧ β and α ∨ β ≤ δ.

The form of this Gröbner basis implies that the standard monomials are
the sortable monomials, those pαpβ · · · pγ with α ≤ β ≤ · · · ≤ γ. Thus the
Hilbert function of Gr,n may be expressed in terms of the combinatorics of
Yr,n. For instance, the dimension of Gr,n is the rank of Yr,n, and its degree
is the number of maximal chains. From Figure 2, these are 6 and 5 for Y2,5,
confirming our previous calculations.

Since the generators g(α, β) are linearly independent, this Gröbner basis
is also a minimal generating set for the ideal. The displayed generator in o51,

p{2,3}p{1,4} − p{1,3}p{2,4} − p{1,2}p{3,4} ,

is g(23, 14), and corresponds to the underlined incomparable pair in Figure 2.
Since there are 5 such incomparable pairs, the Gröbner basis has 5 generators.
As G2,5 has codimension 3, it is not a complete intersection. This shows how

the general enumerative problem from the Schubert calculus gives rise to an
overdetermined system of equations in this global formulation.

The Grassmannian has a useful system of local coordinates given by
Matr,n−r as follows

Y ∈ Matr,n−r 7−→ rowspace [Ir : Y ] ∈ Gr,n . (7)

Let L be a (n−r+1−l)-plane in kn which is the row space of a n−r+1−l
by n matrix, also written L. Then L meets X ∈ Gr,n non-trivially if

maximal minors of
[
L
X

]
= 0 .

Laplace expansion of each minor along the rows of X gives a linear equation
in the Plücker coordinates. In the local coordinates (substituting [Ir : Y ] for
X), we obtain multilinear equations of degree min{r, n− r}. These equations
generate a prime ideal of codimension l.

Suppose each li = 1 in our enumerative problem. Then in the Plücker
coordinates, we have the Plücker ideal of Gr,n together with r(n− r) linear
equations, one for each (n−r)-plane Li. By Theorem 4.1, the Plücker ideal
has a square-free initial ideal, and so the Gröbner homotopy of Example 2.4
may be used to solve this enumerative problem.

Example 4.2. G2,4 ⊂ P5 has equation

p{1,2}p{0,3} − p{1,3}p{0,2} + p{2,3}p{0,1} = 0 . (8)

The condition for H ∈ G2,4 to meet a 2-plane L is the vanishing of

p{1,2}L34 − p{1,3}L24 + p{2,3}L14 + p{1,4}L23 − p{2,4}L13 + p{3,4}L12 , (9)

where Lij is the (i, j)th maximal minor of L.
If l1 = · · · = l4 = 1, we have 5 equations in P5, one quadratic and 4

linear, and so by Bézout’s Theorem there are two 2-planes in k4 that meet 4
general 2-planes non-trivially. This means that there are 2 lines in P3 meeting
4 general lines. In local coordinates, (9) becomes

L34 − L14y11 + L13y12 − L24y21 + L23y22 + L12(y11y22 − y12y21) .

This polynomial has the form of the last specialization in Example 1.2.

4.2 Reality in the Schubert Calculus

Like the other enumerative problems we have discussed, enumerative prob-
lems in the special Schubert calculus are fully real in that all solutions can
be real [22]. That is, given any Schubert data (r, n; l1, . . . , lm), there exist

subspaces L1, . . . , Lm ⊂ Rn such that each of the d(r, n; l1, . . . , lm) r-planes
that meet each Li are themselves real.

This result gives some idea of which choices of the Li give all r-planes
real. Let γ be a fixed rational normal curve in Rn. Then the Li are lin-
ear subspaces osculating γ. More concretely, suppose that γ is the stan-
dard rational normal curve, γ(s) = (1, s, s2, . . . , sn−1). Then the i-plane
Li(s) := 〈γ(s), γ′(s), . . . , γ(i−1)(s)〉 osculating γ at γ(s) is the row space of
the matrix given by oscPlane(i, n, s).

i53 : oscPlane = (i, n, s) -> (
gamma := matrix {toList apply(1..n, i -> s^(i-1))};
L := gamma;
j := 0;
while j < i-1 do (gamma = diff(s, gamma);

L = L || gamma;
j = j+1);

L);

i54 : QQ[s]; oscPlane(3, 6, s)

o55 = | 1 s s2 s3 s4 s5 |
| 0 1 2s 3s2 4s3 5s4 |
| 0 0 2 6s 12s2 20s3 |

3 6
o55 : Matrix QQ [s] <--- QQ [s]

(In o55, the exponents of s are displayed in line: s2 is written s2. Macaulay 2
uses this notational convention to display matrices efficiently.)

Theorem 4.3 ([22]). For any Schubert data (r, n; l1, . . . , lm), there exist
real numbers s1, s2, . . . , sm such that there are d(r, n; l1, . . . , lm) r-planes that
meet each osculating plane Li(si), and all are real.

The inspiration for looking at subspaces osculating the rational normal
curve to study real enumerative geometry for the Schubert calculus is the
following very interesting conjecture of Boris Shapiro and Michael Shapiro,
or more accurately, extensive computer experimentation based upon their
conjecture [20,23,25,30].

Shapiros’s Conjecture. For any Schubert data (r, n; l1, . . . , lm) and for
all real numbers s1, s2, . . . , sm there are d(r, n; l1, . . . , lm) r-planes that meet
each osculating plane Li(si), and all are real.

In addition to Theorem 4.3, (which replaces the quantifier for all by there
exist), the strongest evidence for this Conjecture is the following result of
Eremenko and Gabrielov [10].

Theorem 4.4. Shapiros’s Conjecture is true when either r or n− r is 2.

We test an example of this conjecture for the Schubert data (3, 6; 13, 23),
(where ab is a repeated b times). The algorithms of the Schubert calculus
predict that d(3, 6; 13, 23) = 6. The function spSchub(r, L, P) computes
the ideal of r-planes meeting the row space of L in the Plücker coordinates
Pα.

i56 : spSchub = (r, L, P) -> (
I := ideal apply(subsets(numgens source L,

r + numgens target L), S ->
fold((sum, U) -> sum +
fold((term,i) -> term*(-1)^i, P_(S_U) * det(
submatrix(L, sort toList(set(S) - set(S_U)))), U),

0, subsets(#S, r))));

We are working in the Grassmannian of 3-planes in C6.
i57 : R = QQ[apply(subsets(6,3), i -> p_i )];

The ideal I consists of the special Schubert conditions for the 3-planes to
meet the 3-planes osculating the rational normal curve at the points 1, 2,
and 3, and to also meet the 2-planes osculating at 4, 5, and 6, together
with the Plücker ideal Grassmannian(2, 5, R). Since this is a 1-dimensional
homogeneous ideal, we add the linear form p_{0,1,5} - 1 to make the ideal
zero-dimensional. As before, Grassmannian(2, 5, R) creates the Plücker
ideal of G3,6.

i58 : I = fold((J, i) -> J +
spSchub(3, substitute(oscPlane(3, 6, s), {s=> 1+i}), p) +
spSchub(3, substitute(oscPlane(2, 6, s), {s=> 4+i}), p),
Grassmannian(2, 5, R), {0,1,2}) +

ideal (p_{0,1,5} - 1);

o58 : Ideal of R

This has dimension 0 and degree 6, in agreement with the Schubert calculus.
i59 : dim I, degree I

o59 = (0, 6)

o59 : Sequence

As expected, all roots are real.
i60 : A = R/I; numRealSturm(eliminant(p_{2,3,4}, QQ[Z]))

o61 = 6

There have been many checked instances of this conjecture [23,25,30], and it
has some geometrically interesting generalizations [26].

The question remains for which numbers 0 ≤ d ≤ d(r, n; l1, . . . , lm) do
there exist real planes Li with d(r, n; l1, . . . , lm) r-planes meeting each Li,
and exactly d of them are real. Besides Theorem 4.3 and the obvious parity
condition, nothing is known in general. In every known case, every possibility
occurs—which is not the case in all enumerative problems, even those that
are fully real1. Settling this (for d = 0) has implications for linear systems
theory [20].2

1 For example, of the 12 rational plane cubics containing 8 real points in P2, either
8, 10 or 12 can be real, and there are 8 points with all 12 real [8, Proposition
4.7.3].

2 After this was written, Eremenko and Gabrielov [11] showed that d can be zero
for the enumerative problems given by data (2, 2n, 14n−4) and (2n−2, 2n, 14n−4).

4.3 Transversality in the Schubert Calculus

A basic principle of the classical Schubert calculus is that the intersection
number d(r, n; l1, . . . , lm) has enumerative significance—that is, for general
linear subspaces Li, all solutions appear with multiplicity 1. This basic princi-
ple is not known to hold in general. For fields of characteristic zero, Kleiman’s
Transversality Theorem [15] establishes this principle. When r or n−r is 2,
then Theorem E of [21] establishes this principle in arbitrary characteristic.
We conjecture that this principle holds in general; that is, for arbitrary in-
finite fields and any Schubert data, if the planes Li are in general position,
then the resulting zero-dimensional ideal is radical.

We test this conjecture on the enumerative problem of Section 4.2, which
is not covered by Theorem E of [21]. The function testTransverse(F) tests
transversality for this enumerative problem, for a given field F . It does this
by first computing the ideal of the enumerative problem using random planes
Li.

i62 : randL = (R, n, r, l) ->
matrix table(n-r+1-l, n, (i, j) -> random(0, R));

and the Plücker ideal of the Grassmannian G3,6 Grassmannian(2, 5, R).)
Then it adds a random (inhomogeneous) linear relation 1 + random(1, R)
to make the ideal zero-dimensional for generic Li. When this ideal is zero
dimensional and has degree 6 (the expected degree), it computes the char-
acteristic polynomial g of a generic linear form. If g has no multiple roots, 1
== gcd(g, diff(Z, g)), then the Shape Lemma guarantees that the ideal
was radical. testTransverse exits either when it computes a radical ideal,
or after limit iterations (which is set to 5 for these examples), and prints
the return status.

i63 : testTransverse = F -> (
R := F[apply(subsets(6, 3), i -> q_i )];
continue := true;
j := 0;
limit := 5;
while continue and (j < limit) do (

j = j + 1;
I := fold((J, i) -> J +

spSchub(3, randL(R, 6, 3, 1), q) +
spSchub(3, randL(R, 6, 3, 2), q),
Grassmannian(2, 5, R) +
ideal (1 + random(1, R)),
{0, 1, 2});

if (dim I == 0) and (degree I == 6) then (
lin := promote(random(1, R), (R/I));
g := charPoly(lin, Z);
continue = not(1 == gcd(g, diff(Z, g)));
));

if continue then << "Failed for the prime " << char F <<
" with " << j << " iterations" << endl;

if not continue then << "Succeeded for the prime " <<
char F << " in " << j << " iteration(s)" << endl;

);

Since 5 iterations do not show transversality for F2,
i64 : testTransverse(ZZ/2);
Failed for the prime 2 with 5 iterations

we can test transversality in characteristic 2 using the field with four elements,
F4 = GF 4.

i65 : testTransverse(GF 4);
Succeeded for the prime 2 in 3 iteration(s)

We do find transversality for F7.
i66 : testTransverse(ZZ/7);
Succeeded for the prime 7 in 2 iteration(s)

We have tested transversality for all primes less than 100 in every enumer-
ative problem involving Schubert conditions on 3-planes in k6. These include
the problem above as well as the problem of 42 3-planes meeting 9 general
3-planes.3

5 The 12 Lines: Reprise

The enumerative problems of Section 3 were formulated in local coordi-
nates (7) for the Grassmannian of lines in P3 (Grassmannian of 2-dimensional
subspaces in k4). When we formulate the problem of Section 3.2 in the global
Plücker coordinates of Section 4.1, we find some interesting phenomena. We
also consider some related enumerative problems.

5.1 Global Formulation

A quadratic form q on a vector space V over a field k not of characteristic 2
is given by q(u) = (ϕ(u),u), where ϕ : V → V ∗ is a symmetric linear map,
that is (ϕ(u),v) = (ϕ(v),u). Here, V ∗ is the linear dual of V and ( · , · ) is
the pairing V ⊗V ∗ → k. The map ϕ induces a quadratic form ∧rq on the rth
exterior power ∧rV of V through the symmetric map ∧rϕ : ∧r V → ∧rV ∗ =
(∧rV )∗. The action of ∧rV ∗ on ∧rV is given by

(x1 ∧ x2 ∧ · · · ∧ xr, y1 ∧ y2 ∧ · · · ∧ yr) = det |(xi,yj)| , (10)

where xi ∈ V ∗ and yj ∈ V .
When we fix isomorphisms V ' kn ' V ∗, the map ϕ is given by a

symmetric n × n matrix M as in Section 3.2. Suppose r = 2. Then for
u,v ∈ kn,

∧2q(u ∧ v) = det
[

uMut uMvt

vMut vMvt

]
,

which is Equation (6) of Section 3.2.
3 After this was written, we discovered an elementary proof of transversality for

the enumerative problems given by data (r, n; 1r(n−r)), where the conditions are
all codimension 1 [24].

Proposition 5.1. A line ` is tangent to a quadric V(q) in Pn−1 if and only
if its Plücker coordinate ∧2` ∈ P(n2)−1 lies on the quadric V(∧2q).

Thus the Plücker coordinates for the set of lines tangent to 4 general
quadrics in P3 satisfy 5 quadratic equations: The single Plücker relation (8)
together with one quadratic equation for each quadric. Thus we expect the
Bézout number of 25 = 32 such lines. We check this.

The procedure randomSymmetricMatrix(R, n) generates a random sym-
metric n× n matrix with entries in the base ring of R.

i67 : randomSymmetricMatrix = (R, n) -> (
entries := new MutableHashTable;
scan(0..n-1, i -> scan(i..n-1, j ->

entries#(i, j) = random(0, R)));
matrix table(n, n, (i, j) -> if i > j then

entries#(j, i) else entries#(i, j))
);

The procedure tangentEquation(r, R, M) gives the equation in Plücker
coordinates for a point in P(nr)−1 to be isotropic with respect to the bilinear
form ∧rM (R is assumed to be the coordinate ring of P(nr)−1). This is the
equation for an r-plane to be tangent to the quadric associated to M .

i68 : tangentEquation = (r, R, M) -> (
g := matrix {gens(R)};
(entries(g * exteriorPower(r, M) * transpose g))_0_0
);

We construct the ideal of lines tangent to 4 general quadrics in P3.
i69 : R = QQ[apply(subsets(4, 2), i -> p_i )];

i70 : I = Grassmannian(1, 3, R) + ideal apply(0..3, i ->
tangentEquation(2, R, randomSymmetricMatrix(R, 4)));

o70 : Ideal of R

As expected, this ideal has dimension 0 and degree 32.
i71 : dim Proj(R/I), degree I

o71 = (0, 32)

o71 : Sequence

5.2 Lines Tangent to 4 Spheres

That calculation raises the following question: In Section 3.2, why did we
obtain only 12 lines tangent to 4 spheres? To investigate this, we generate
the global ideal of lines tangent to the spheres of Section 3.2.

i72 : I = Grassmannian(1, 3, R) +
ideal (tangentEquation(2, R, Sphere(0,0,0,5)),

tangentEquation(2, R, Sphere(4,1,1,5)),
tangentEquation(2, R, Sphere(1,4,1,5)),
tangentEquation(2, R, Sphere(1,1,4,5)));

o72 : Ideal of R

We compute the dimension and degree of V(I).
i73 : dim Proj(R/I), degree I

o73 = (1, 4)

o73 : Sequence

The ideal is not zero dimensional; there is an extraneous one-dimensional
component of zeroes with degree 4. Since we found 12 lines in Section 3.2
using the local coordinates (7), the extraneous component must lie in the
complement of that coordinate patch, which is defined by the vanishing of
the first Plücker coordinate, p{0,1}. We saturate I by p{0,1} to obtain the
desired lines.

i74 : Lines = saturate(I, ideal (p_{0,1}));

o74 : Ideal of R

This ideal does have dimension 0 and degree 12, so we have recovered the
zeroes of Section 3.2.

i75 : dim Proj(R/Lines), degree(Lines)

o75 = (0, 12)

o75 : Sequence

We investigate the rest of the zeroes, which we obtain by taking the ideal
quotient of I and the ideal of lines. As computed above, this has dimension
1 and degree 4.

i76 : Junk = I : Lines;

o76 : Ideal of R

i77 : dim Proj(R/Junk), degree Junk

o77 = (1, 4)

o77 : Sequence

We find the support of this extraneous component by taking its radical.
i78 : radical(Junk)

2 2 2
o78 = ideal (p , p , p , p + p + p )

{0, 3} {0, 2} {0, 1} {1, 2} {1, 3} {2, 3}

o78 : Ideal of R

From this, we see that the extraneous component is supported on an imagi-
nary conic in the P2 of lines at infinity.

To understand the geometry behind this computation, observe that the
sphere with radius r and center (a, b, c) has homogeneous equation

(x− wa)2 + (y − wb)2 + (z − wc)2 = r2w2 .

At infinity, w = 0, this has equation

x2 + y2 + z2 = 0 .

The extraneous component is supported on the set of tangent lines to this
imaginary conic. Aluffi and Fulton [1] studied this problem, using geometry
to identify the extraneous ideal and the excess intersection formula [13] to
obtain the answer of 12. Their techniques show that there will be 12 isolated
lines tangent to 4 quadrics which have a smooth conic in common.

When the quadrics are spheres, the conic is the imaginary conic at infinity.
Fulton asked the following question: Can all 12 lines be real if the (real) four
quadrics share a real conic? We answer his question in the affirmative in the
next section.

5.3 Lines Tangent to Real Quadrics Sharing a Real Conic

We consider four quadrics in P3
R

sharing a non-singular conic, which we will
take to be at infinity so that we may use local coordinates for G2,4 in our
computations. The variety V(q) ⊂ P3

R
of a nondegenerate quadratic form q

is determined up to isomorphism by the absolute value of the signature σ of
the associated bilinear form. Thus there are three possibilities, 0, 2, or 4, for
|σ|.

When |σ| = 4, the real quadric V(q) is empty. The associated symmetric
matrix M is conjugate to the identity matrix, so ∧2M is also conjugate to
the identity matrix. Hence V(∧2q) contains no real points. Thus we need not
consider quadrics with |σ| = 4.

When |σ| = 2, we have V(q) ' S2, the 2-sphere. If the conic at infinity
is imaginary, then V(q) ⊂ R3 is an ellipsoid. If the conic at infinity is real,
then V(q) ⊂ R3 is a hyperboloid of two sheets. When σ = 0, we have V(q) '
S1 × S1, a torus. In this case, V(q) ⊂ R3 is a hyperboloid of one sheet and
the conic at infinity is real.

Thus either we have 4 ellipsoids sharing an imaginary conic at infinity,
which we studied in Section 3.2; or else we have four hyperboloids sharing a
real conic at infinity, and there are five possible combinations of hyperboloids
of one or two sheets sharing a real conic at infinity. This gives six topologically
distinct possibilities in all.

Theorem 5.2. For each of the six topologically distinct possibilities of four
real quadrics sharing a smooth conic at infinity, there exist four quadrics
having the property that each of the 12 lines in C3 simultaneously tangent to
the four quadrics is real.

Proof. By the computation in Section 3.2, we need only check the five possi-
bilities for hyperboloids. We fix the conic at infinity to be x2 + y2 − z2 = 0.
Then the general hyperboloid of two sheets containing this conic has equation
in R3

(x− a)2 + (y − b)2 − (z − c)2 + r = 0 , (11)

(with r > 0). The command Two(a,b,c,r) generates the associated sym-
metric matrix.



From Enumerative Geometry to Solving Equations 25

i79 : Two = (a, b, c, r) -> (
matrix{{a^2 + b^2 - c^2 + r ,-a ,-b , c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ c , 0 , 0 ,-1 }}

);

The general hyperboloid of one sheet containing the conic x2 + y2 − z2 = 0
at infinity has equation in R3

(x− a)2 + (y − b)2 − (z − c)2 − r = 0 , (12)

(with r > 0). The command One(a,b,c,r) generates the associated sym-
metric matrix.

i80 : One = (a, b, c, r) -> (
matrix{{a^2 + b^2 - c^2 - r ,-a ,-b , c },

{ -a , 1 , 0 , 0 },
{ -b , 0 , 1 , 0 },
{ c , 0 , 0 ,-1 }}

);

We consider i quadrics of two sheets (11) and 4 − i quadrics of one
sheet (12). For each of these cases, the table below displays four 4-tuples
of data (a, b, c, r) which give 12 real lines. (The data for the hyperboloids of
one sheet are listed first.)

i Data
0 (5, 3, 3, 16), (5,−4, 2, 1), (−3,−1, 1, 1), (2,−7, 0, 1)
1 (3,−2,−3, 6), (−3,−7,−6, 7), (−6, 3,−5, 2), (1, 6,−2, 5)
2 (6, 4, 6, 4), (−1, 3, 3, 6), (−7,−2, 3, 3), (−6, 7,−2, 5)
3 (−1,−4,−1, 1), (−3, 3,−1, 1), (−7, 6, 2, 9), (5, 6,−1, 12)
4 (5, 2,−1, 25), (6,−6, 2, 25), (−7, 1, 6, 1), (3, 1, 0, 1)

We test each of these, using the formulation in local coordinates of Sec-
tion 3.2.

i81 : R = QQ[y11, y12, y21, y22];

i82 : I = ideal (tangentTo(One( 5, 3, 3,16)),
tangentTo(One( 5,-4, 2, 1)),
tangentTo(One(-3,-1, 1, 1)),
tangentTo(One( 2,-7, 0, 1)));

o82 : Ideal of R

i83 : numRealSturm(charPoly(promote(y22, R/I), Z))

o83 = 12

i84 : I = ideal (tangentTo(One( 3,-2,-3, 6)),
tangentTo(One(-3,-7,-6, 7)),
tangentTo(One(-6, 3,-5, 2)),
tangentTo(Two( 1, 6,-2, 5)));

o84 : Ideal of R

i85 : numRealSturm(charPoly(promote(y22, R/I), Z))

o85 = 12

i86 : I = ideal (tangentTo(One( 6, 4, 6, 4)),
tangentTo(One(-1, 3, 3, 6)),
tangentTo(Two(-7,-2, 3, 3)),
tangentTo(Two(-6, 7,-2, 5)));

o86 : Ideal of R

i87 : numRealSturm(charPoly(promote(y22, R/I), Z))

o87 = 12

i88 : I = ideal (tangentTo(One(-1,-4,-1, 1)),
tangentTo(Two(-3, 3,-1, 1)),
tangentTo(Two(-7, 6, 2, 9)),
tangentTo(Two( 5, 6,-1,12)));

o88 : Ideal of R

i89 : numRealSturm(charPoly(promote(y22, R/I), Z))

o89 = 12

i90 : I = ideal (tangentTo(Two( 5, 2,-1,25)),
tangentTo(Two( 6,-6, 2,25)),
tangentTo(Two(-7, 1, 6, 1)),
tangentTo(Two( 3, 1, 0, 1)));

o90 : Ideal of R

i91 : numRealSturm(charPoly(promote(y22, R/I), Z))

o91 = 12

ut

In each of these enumerative problems there are 12 complex solutions. For
each, we have done other computations showing that every possible number
of real solutions (0, 2, 4, 6, 8, 10, or 12) can occur.

5.4 Generalization to Higher Dimensions

We consider lines tangent to quadrics in higher dimensions. First, we rein-
terpret the action of ∧rV ∗ on ∧rV described in (10) as follows. The vec-
tors x1, . . . ,xr and y1, . . . ,yr define maps α : kr → V ∗ and β : kr → V .
The matrix [(xi, yj)] is the matrix of the bilinear form on kr given by
〈u, v〉 := (α(u), β(v)). Thus (10) vanishes when the bilinear form 〈 · , · 〉
on kr is degenerate.

Now suppose that we have a quadratic form q on V given by a symmetric
map ϕ : V → V ∗. This induces a quadratic form and hence a quadric on
any r-plane H in V (with H 6⊂ V(q)). This induced quadric is singular when
H is tangent to V(q). Since a quadratic form is degenerate only when the
associated projective quadric is singular, we see that H is tangent to the
quadric V(q) if and only if (∧rϕ(∧rH), ∧rH) = 0. (This includes the case
H ⊂ V(q).) We summarize this argument.

Theorem 5.3. Let ϕ : V → V ∗ be a linear map with resulting bilinear form
(ϕ(u), v). Then the locus of r-planes in V for which the restriction of this
form is degenerate is the set of r-planes H whose Plücker coordinates are
isotropic, (∧rϕ(∧rH), ∧rH) = 0, with respect to the induced form on ∧rV .

When ϕ is symmetric, this is the locus of r-planes tangent to the associated
quadric in P(V ).

We explore the problem of lines tangent to quadrics in Pn. From the
calculations of Section 5.1, we do not expect this to be interesting if the
quadrics are general. (This is borne out for P4: we find 320 lines in P4 tangent
to 6 general quadrics. This is the Bézout number, as deg G2,5 = 5 and the
condition to be tangent to a quadric has degree 2.) This problem is interesting
if the quadrics in Pn share a quadric in a Pn−1. We propose studying such
enumerative problems, both determining the number of solutions for general
such quadrics, and investigating whether or not it is possible to have all
solutions be real.

We use Macaulay 2 to compute the expected number of solutions to this
problem when r = 2 and n = 4. We first define some functions for this
computation, which will involve counting the degree of the ideal of lines
in P4 tangent to 6 general spheres. Here, X gives local coordinates for the
Grassmannian, M is a symmetric matrix, tanQuad gives the equation in X
for the lines tangent to the quadric given by M .

i92 : tanQuad = (M, X) -> (
u := X^{0};
v := X^{1};
(u * M * transpose v)^2 -
(u * M * transpose u) * (v * M * transpose v)
);

nSphere gives the matrix M for a sphere with center V and squared radius
r, and V and r give random data for a sphere.

i93 : nSphere = (V, r) ->
(matrix {{r + V * transpose V}} || transpose V ) |
( V || id_((ring r)^n)
);

i94 : V = () -> matrix table(1, n, (i,j) -> random(0, R));

i95 : r = () -> random(0, R);

We construct the ambient ring, local coordinates, and the ideal of the enu-
merative problem of lines in P4 tangent to 6 random spheres.

i96 : n = 4;

i97 : R = ZZ/1009[flatten(table(2, n-1, (i,j) -> z_(i,j)))];

i98 : X = 1 | matrix table(2, n-1, (i,j) -> z_(i,j))

o98 = | 1 0 z_(0,0) z_(0,1) z_(0,2) |
| 0 1 z_(1,0) z_(1,1) z_(1,2) |

2 5
o98 : Matrix R <--- R

i99 : I = ideal (apply(1..(2*n-2),
i -> tanQuad(nSphere(V(), r()), X)));

o99 : Ideal of R

We find there are 24 lines in P4 tangent to 6 general spheres.
i100 : dim I, degree I

o100 = (0, 24)

o100 : Sequence

The expected numbers of solutions we have obtained in this way are displayed
in the table below. The numbers in boldface are those which are proven.4

n 2 3 4 5 6
# expected 4 12 24 48 96

Acknowledgments. We thank Dan Grayson and Bernd Sturmfels: some of the
procedures in this chapter were written by Dan Grayson and the calculation in
Section 5.2 is due to Bernd Sturmfels.

Bézout number 1, 22,
27

